module mod_modulo_util
  implicit none
  integer(8), parameter :: md = 1000000007_8
  integer(8), allocatable :: f(:), invf(:)
contains
  subroutine init(n)
    integer, intent(in) :: n
    integer :: i
    if (allocated(f)) deallocate(f)
    if (allocated(invf)) deallocate(invf)
    allocate(f(0:n),invf(0:n))
    f = 0_8
    invf = 0_8
    f(0) = 1_8
    do i = 1, n
      f(i) = mod(f(i-1)*int(i,8),md)
    end do
    invf(n) = inv(f(n))
    do i = n, 1, -1
      invf(i-1) = mod(invf(i)*int(i,8),md)
    end do
  end
  integer(8) function perm(n,k)
    integer, intent(in) :: n, k
    perm = 0_8
    if (k > n .or. n < 0 .or. k < 0) return
    perm = mod(f(n)*invf(n-k),md)
  end
  integer(8) function comb(n,k)
    integer, intent(in) :: n, k
    comb = mod(perm(n,k)*invf(k),md)
  end
  integer(8) function homo(n,k)
    integer, intent(in) :: n, k
    homo = comb(n+k-1,k)
  end
  function inv(n) result(y)
    integer(8), intent(in) :: n
    integer(8) :: a, b, x, y, t, q
    a = mod(n,md)
    b = md
    x = 0_8
    y = 1_8
    do while (b /= 0_8)
      q = a/b
      t = b
      b = mod(a,b)
      a = t
      t = y
      y = x
      x = t-q*x
    end do
    y = mod(mod(y,md)+md,md)
  end
  function pow(a,b) result(r)
    integer(8), intent(in) :: a
    integer, intent(in) :: b
    integer :: i
    integer(8) :: r, p
    r = 1_8
    p = a
    do i = 0, 32
      if (btest(b,i)) r = mod(r*p,md)
      p = mod(p*p,md)
    end do
  end
  integer(8) function fact(n)
    integer, intent(in) :: n
    fact = 0_8
    if (n < 0) return
    fact = f(n)
  end
  integer(8) function comb3(a,b,c,d)
    integer, intent(in) :: a, b, c, d
    comb3 = 0_8
    if (a < 0 .or. b < 0 .or. c < 0 .or. d < 0 .or. a < b+c+d) return
    comb3 = mod(mod(invf(b)*invf(c),md)*mod(invf(d)*invf(a-b-c-d),md),md)
    comb3 = mod(comb3*f(a),md)
  end
end module mod_modulo_util
program one_two_three_pairs
  use mod_modulo_util
  implicit none
  integer(8) :: ans = 0_8
  integer :: n, a, b, c, p, q, r
  read(*,*) n, a, b, c
  if (mod(b,2) == 1) then
    write(*,'(i0)') 0
    stop
  end if
  call init(n)
  do p = 0, min(a,c)
    do q = mod(c-p,3), c-p, 3
      r = (c-p-q)/3
      ans = mod(ans+mod(comb3(r+a+b/2,r,b/2,a-p)*homo(b/2,q),md),md)
    end do
  end do
  write(*,'(i0)') ans
end program one_two_three_pairs